
Элемент плоской фермы. 

 

Фермой называется стержневая система, остающаяся геометрически неизменяемой 

после условной замены её жестких узлов шарнирными.  

Рассмотрим элемент, с помощью которого можно рассчитывать фермы. Этот элемент 

имеет произвольную ориентацию в плоскости 

Зная координаты 
( , ),( , )x y x y1 1 2 2 концов 1 и 2 стержня, можно определить его 

длину l  и направляющие косинусы c, s: 

l x x y y

c x x l

s y y l

   

  

  

( ) ( )

cos ( )

sin ( )

2 1

2

2 1

2

2 1

2 1




                                                                (14)      

Перемещения стержня определяется составляющими 
( , ), ( , )u v u v1 1 2 2  перемещений 

его концов. Сгруппируем их в один вектор столбец  

Удлиннение стержня равно проекции разности перемещений его концов на ось 

стержня 

Если модуль упругости обозначим через E, а поперечное сечение - через F, то на 



основе закона Гука получим растягивающую силу 

Решения задачи при отыскании узловых значений перемещений можно получить 

путем определения минимума функции потенциальной энергии. Потенциальная энергия 

системы может быть представлена как сумму потенциальных энергий каждого элемента т.е. 

Полная потенциальная энергия конечного элемента П i  может быть представлена в 

виде 

где V- энергия деформации, А- работа узловых сил. 

Энергия деформации равна 

Работа узловых сил равна 

A P u { } { }

Чтобы минимизировать величину П, продифференциируем её по {u} и приравняем 

результат нулю. 

Окончательно будем иметь 

Для всей системы  

{ }P
- глобальный вектор нагрузок 

n - число конечных элементов. 

 



Задача. Определить перемещения 2 узла фермы показанной на чертеже. 

     Определим координаты узлов 

 1 2 3 

x 0.0 4.0 8.0 

y 0.0 4.0 0.0 

Длины 1 и 2 элемента соответственно равны 

а косинусы и синусы элементов соответственно равны 

Матрицы жесткости каждого элемента равны 

Просуммируем их 

Алгоритм составления глобальной матрицы состоит из следующего. Для каждого 

элемента составляется вектор коэффициентов.  

Например для первого элемента  

1 2 3 4 , 

а для второго 

3 4 5 6 

Порядковый номер в векторе показывает номер члена в локальной матрице, а его 

значение показывает номер в глобальной матрице.  

Приступим к учёту граничных условий. Из условия задачи 1 и 2, а также 5 и 6 степени 

свободы закреплены, т.е перемещения 1,2,5,6 равны нулю. Поэтому в системе уравнений 

обнуляются элементы соответствующие этим степеням свободы.  

Чтобы в программах не делать удаление строк и столбцов соответствующие 

закрепленным степеням свободы, нужно нумеровать только неизвестные перемещения. Т.е.  

для первого элемента  

0 0 1 2 , 

а для второго 

1 2 0 0 

Вектор нагрузок будет равен 

{ }F 








5

0
 

Система линейных уравнений и ее решение имеет вид 

ПРОГРАММА РАСЧЕТА ПЕРЕМЕЩЕНИЙ 

10 DIM K(4, 4)    'массив для матрицы [К] 

20 DIM ST(4)      'массив для списка индексов [S] 

30 READ N, NE     'ввод числа уравнений и элементов 

40 DIM A(N, N)    'массив для матрицы [A] 

50 DIM X(2) 

60 DIM Y(2) 

70 FOR J = 1 TO NE    'цикл по элементам 

80 REM ввод списка индексов матрицы 

90 REM (адреса матрицы [K] в матрице [A]) 

100 READ ST(1), ST(2), ST(3), ST(4) 

110 READ X(1), Y(1), X(2), Y(2) 

120 READ E, A 

130 REM ввод модуля упругости и площади 

140 DX = X(2) - X(1): DY = Y(2) - Y(1) 

150 EL = SQR(DX * DX + DY * DY) 



160 EL = 1 / EL: SF = E * A * EL 

170 REM построение матрицы жесткости элемента 

180 C = DX * EL: S = DY * EL 

190 CC = C * C: SS = S * S: CS = C * S 

200 REM матрица жесткости 

210 K(1, 1) = CC:  K(1, 2) = CS:  K(1, 3) = -CC: K(1, 4) = -CS 

220 K(2, 1) = CS:  K(2, 2) = SS:  K(2, 3) = -CS: K(2, 4) = -SS 

230 K(3, 1) = -CC: K(3, 2) = -CS: K(3, 3) = CC:  K(3, 4) = CS 

240 K(4, 1) = -CS: K(4, 2) = -SS: K(4, 3) = CS:  K(4, 4) = SS 

250 REM сборка матрицы [A] 

260   FOR L = 1 TO 4 

270    FOR M = 1 TO 4 

280     A(ST(L), ST(M)) = A(ST(L), ST(M)) + K(L, M) * SF 

290    NEXT M 

300   NEXT L 

310 NEXT J 

320 REM Ввод нагрузок 

330 DIM B(N)          'массив для матрицы нагрузок 

340 FOR J = 1 TO N 

350  READ B(J)        'ввод матрицы нагрузок 

360 NEXT J 

370 REM Решение системы линейных уравнений 

380 DIM U(N)  'массив для неизвестных перемещений 

390 'ПРЯМОЙ ХОД 

400 FOR J = 1 TO N - 1 

410  FOR JJ = J + 1 TO N 

420   XM = A(J, JJ) / A(J, J) 

430   FOR I = JJ TO N 

440    A(JJ, I) = A(JJ, I) - A(J, I) * XM 

450   NEXT I 

460   B(JJ) = B(JJ) - B(J) * XM 

470  NEXT JJ 

480 NEXT J 

490 'ОБРАТРЫЙ ХОД 

500 U(N) = B(N) / A(N, N) 

510 FOR J = N - 1 TO 1 STEP -1 

520  SUM = 0 

530  FOR I = J + 1 TO N 

540   SUM = SUM + A(J, I) * U(I) 

550  NEXT I 

560  U(J) = (B(J) - SUM) / A(J, J) 

570 NEXT J 

580 '**** ВЫВОД ПЕРЕМЕЩЕНИЙ **** 

590 FOR I = 1 TO N 

600  PRINT USING "U(##)=##.#########"; I; U(I) 

610 NEXT I 

620 END 

630 REM Исходные данные к примеру. 

640 DATA 2,2 

650 DATA 0,0,1,2 

660 DATA 0,0,4,4 

670 DATA 1,1 

680 DATA 1,2,0,0 

690 DATA 4,4,8,0 

700 DATA 1,1 

710 REM элементы матрицы нагрузок 

720 DATA 5,0 


